
Introduction 

To illustrate the range of frequency sweeping possible with the DRE, this application note 
provides an effect similar to placing a 45 rpm phonograph record - the kind with the large hole - 
off-center by the greatest amount possible on a turntable with a standard 0.285" spindle. 
 
 

Algorithm 

The hole on a 45 rpm record is 1.5" in diameter.  If the record is moved off center so that the 
edge of the hole touches the spindle, the record will be 0.5*(1.5"-0.285") = 0.6075" off-center.  A 
typical distance from the center of the record to the stylus is 2.875" (about 1/3 of the way 
through a song that fills up as much of the record as possible).  Approximating the geometric 
functions involved, the stylus position will be as much as 0.6075" ahead or behind of where it 
would be if the record were properly centered. 

 
 
To figure out the time deviation, we first need to calculate the deviation distance.  In the figure 
above, the two dashed lines show the maximum variation of the stylus position, at 0.6075".  The 
three thin lines are the radius line to the maximum positive, maximum negative, and zero 
variation points, with the typical distance of 2.875".   The deviation distance may be determined 
as follows: 

⇒ The center and one outer edge thin line, plus its connecting dashed line, form 

an equilateral triangle.  By bisecting the corner angle β with the thick line, we 

may calculate the half angle α thus formed with the formula sin(α) = length of 
opposite side / length of hypotenuse.  To simplify the math, we use the 

approximation that sin(α) ≈ α for small angles. 

  β = 2*α ≈ 2*sin(α) = 2*(0.6075"/2)/2.875" = 0.2113 radians 

  0.2113 radians / 2π radians in a circle = 0.03363 of the circumference 

⇒ Using the distance, plus the speed of the record (45rpm, or 0.75rps), the time 
deviation may be determined. 

  0.03363/0.75rps = 44.84ms 

At 48000 samples/second, 44.84ms is 2152.33 samples ahead or behind.  For a stereo record, 
the right channel is the outer wall of the groove and is about 1 mil further from the center of the 
record than the left channel.  Allowing the previous result to be the left channel, the variations 
for the right channel are a little smaller, 2151.58 samples. 

The output of a magnetic phonograph cartridge is proportional to velocity, so as the sound 
pitches up the amplitude will rise.  However, this is mostly compensated for by the RIAA 
playback response, which declines with rising frequency.  Thus for a gross effect such as this, 
there's no need to attempt to change the amplitude. 
 
The severely misaligned stylus will produce distortion, this also is not simulated. 
    
The running-ahead of the audio will be duplicated by pitching up the audio and the running-
behind by pitching down.  Pitching up results in the possibility of aliasing.  To prevent this, the 
program includes an anti-aliasing filter with a cutoff frequency of approximately 19kHz.  The 
filter has only moderate out-of-band rejection; a better filter could be designed and implemented 
in the DRE, with some difficulty, as the limited coefficient range of the DRE makes implementing 
high quality filters difficult.  The strength of the DRE is in its high quality, wide range digital 
oscillators and the functions that can be derived from them, thus the filter consumes most of the 
code, and the varying pitch shift only a small part. 

The frequency coefficient for the oscillators is 26, implying a 45.46 rpm rate for the "record", 
which is as close as the DRE can come to 45 rpm with a sample rate of 48000 per second.  The 
left channel amplitude is 2152.33 x 8 ≈ 17219, the right channel 2150.58 x 8 ≈17213. 
 
 

Source Code 

LFO0 SIN AMP=17219 FREQ=26  ;LEFT, 45.46 RPM 
LFO1 SIN AMP=17213 FREQ=26  ;RIGHT 
 
MEM BUFL 17  ;left channel input-FIR buffer 
MEM SWMEML 4400  ;left channel sweep memory 
MEM BUFR 17 
MEM SWMEMR 4400 
 
RZP ADCL  K=127  ;read left channel input 
WZP BUFL  K=1  ;write input to FIR buffer, first FIR coefficient 
RAP BUFL+1  K=-2  ;second FIR stage 
RAP BUFL+2 K=2 
RAP BUFL+3 K=-1 
RAP BUFL+4 K=-3 
RAP BUFL+5 K=10 
RAP BUFL+6 K=-18 
RAP BUFL+7 K=25 
RAP BUFL+8 K=100 
RAP BUFL+9 K=25 
RAP BUFL+10 K=-18 
RAP BUFL+11 K=10 
RAP BUFL+12 K=-3 
RAP BUFL+13 K=-1 
RAP BUFL+14 K=2 
RAP BUFL+15 K=-2 
RAP BUFL+16 K=1 
WZP SWMEML K=0  ;save FIR output to beginning of sweep memory 
CHR0 RZP SWMEML"  +SIN  LATCH  COMPK   ;get data at swept memory address, scale 
CHR0 RAP SWMEML"+1  +SIN    ;add (1-scale)x(data) at next address 
WAP OUTL  K=0   ;write result to left output 

RZP ADCR  K=127  ;read right channel input 
WZP BUFR  K=1  ;write input to FIR buffer, first FIR coefficient 
RAP BUFR+1  K=-2  ;second FIR stage 
RAP BUFR+2 K=2 
RAP BUFR+3 K=-1 
RAP BUFR+4 K=-3 
RAP BUFR+5 K=10 
RAP BUFR+6 K=-18 
RAP BUFR+7 K=25 
RAP BUFR+8 K=100 
RAP BUFR+9 K=25 
RAP BUFR+10 K=-18 
RAP BUFR+11 K=10 
RAP BUFR+12 K=-3 
RAP BUFR+13 K=-1 
RAP BUFR+14 K=2 
RAP BUFR+15 K=-2 
RAP BUFR+16  K=1 
WZP SWMEMR  K=0  ;save FIR output to beginning of sweep memory 
CHR1 RZP SWMEMR"  LATCH  +SIN  COMPK   ;get data at swept memory address, scale 
CHR1 RAP SWMEMR"+1  +SIN    ;add (1-scale)x(data) at next address 
WAP OUTR K=0   ;write result to right output 
 
RZP BUFR+0x40 K=0  ;refresh 
RZP BUFR+0x80 K=0  ;refresh 
RZP BUFR+0xC0 K=0  ;refresh 
RZP BUFR+0x100 K=0  ;refresh 
RZP BUFR+0x140 K=0  ;refresh 
RZP BUFR+0x180 K=0  ;refresh 
RZP BUFR+0x1C0 K=0  ;refresh 
RZP BUFR+0x200 K=0  ;refresh 
RZP BUFR+0x240 K=0  ;refresh 
RZP BUFR+0x280 K=0  ;refresh 
RZP BUFR+0x2C0 K=0  ;refresh 
RZP BUFR+0x300 K=0  ;refresh 
RZP BUFR+0x340 K=0  ;refresh 
RZP BUFR+0x380 K=0  ;refresh 
RZP BUFR+0x3C0 K=0  ;last refresh, END OF PROGRAM

NOTICE 

Wavefront Semiconductor reserves the right to make changes to their products 
or to discontinue any product or service without notice.  All products are sold 
subject to terms and conditions of sale supplied at the time of order 
acknowledgement.  Wavefront Semiconductor assumes no responsibility for the 
use of any circuits described herein, conveys no license under any patent or 
other right, and makes no representation that the circuits are free of patent 
infringement.  Information contained herein is only for illustration purposes and 
may vary depending upon a user’s specific application.  While the information in 
this publication has been carefully checked, no responsibility is assumed for 
inaccuracies. 

Wavefront Semiconductor products are not designed for use in applications 
which involve potential risks of death, personal injury, or severe property or 
environmental damage or life support applications where the failure or 
malfunction of the product can reasonably be expected to cause failure of the life 
support system or to significantly affect its safety or effectiveness. 

All trademarks and registered trademarks are property of their respective owners. 

